Solutions to Problems on the
Gram-Schmidt Process

(I) The matrix A to be considered here is:

$$A := \begin{bmatrix} 1 & 2 & 4 & 1 \\ 0 & 2 & 2 & 0 \\ 2 & 3 & 7 & 1 \\ 1 & 1 & 3 & 0 \end{bmatrix}$$

Let's label the columns as $v1$, $v2$, $v3$ and $v4$, and write them as row vectors. You would enter these vectors into Maple as follows:

```
> v1 := vector([1,0,2,1]);
```

$$v1 := [1, 0, 2, 1]$$

```
> v2 := vector([2,2,3,1]);
```

$$v2 := [2, 2, 3, 1]$$

```
> v3 := vector([4,2,7,3]);
```

$$v3 := [4, 2, 7, 3]$$

```
> v4 := vector([1,0,1,0]);
```

$$v4 := [1, 0, 1, 0]$$

Now let's begin the process, which we do by normalizing $v1$. To do this, we simply divide $v1$ by its length. This gets done a lot, so there is a Maple command for this purpose,called *normalize( )* naturally enough:

```
> u1 := normalize(v1);
```

$$u1 := \left[\frac{1}{6}\sqrt{6},\ 0,\ \frac{1}{3}\sqrt{6},\ \frac{1}{6}\sqrt{6}\right]$$

Now we form $w2$. To do this in Maple, use the *dotprod(,)* command to compute the dot product, and don't forget the $*$ for multiplication.

```
> w2 := v2 - dotprod(u1,v2)*u1;
```

$$w2 := v2 - \frac{3}{2}\sqrt{6}\,u1$$

To see the numerical values of the components, use *evalm( )*, the matrix evaluation command.

```
> evalm(w2);
```

$$\left[\frac{1}{2},\ 2,\ 0,\ \frac{-1}{2}\right]$$

Since $w2$ is not the zero vector, we can normalize it. Doing so gives us $u2$:

```
> u2 := normalize(w2);
```

$$u2 := \left[\frac{1}{6}\sqrt{2},\ \frac{2}{3}\sqrt{2},\ 0,\ -\frac{1}{6}\sqrt{2}\right]$$

Now we compute $w3$:

```
> w3 := v3 - dotprod(u1,v3)*u1 - dotprod(u2,v3)*u2;
```

$$w3 := v3 - \frac{7}{2}\sqrt{6}\,u1 - \frac{3}{2}\sqrt{2}\,u2$$

```
> evalm(w3);
```

$$[0, 0, 0, 0]$$

Since $w3$ is the zero vector, $v3$ is a linear combination of $u1$ and $u2$, and we skip on to $v4$ to see if there will be a $u3$ in our basis or not. In these solutions, we will keep calling $v4$ by its original name, instead of renaming it $v3$.

```
> w3 := v4 - dotprod(u1,v4)*u1 - dotprod(u2,v4)*u2;
```

$$w3 := v4 - \frac{1}{2}\sqrt{6}\,u1 - \frac{1}{6}\sqrt{2}\,u2$$

```
> evalm(w3);
```

$$\left[\frac{4}{9}, \frac{-2}{9}, 0, \frac{-4}{9}\right]$$

This is not the zero vector, so we can normalize to obtain $u3$:

```
> u3 := normalize(w3);
```

$$u3 := \left[\frac{2}{3}, \frac{-1}{3}, 0, \frac{-2}{3}\right]$$

Now, $u1$, $u2$ and $u3$ are our orthonormal basis for the column space of A, at least when we rewrite them as column vectors. The problem didn't actually ask you to do the next step, but let's do one more thing: Let's find a fourth vector $u4$ so that $u1$, $u2$, $u3$ and $u4$ are an orthonormal basis of R^4. To do this, continue with the Gram–Schmidt process using the standard basis vectors $e1$, $e2$, $e3$ and $e4$ as the next vectors in the list of vectors to be orthonormalized. Since these span R^4, we are sure to complete our basis using them. Also, our basis is complete as soon as we find $u4$, so we don't necessarily need to use all of the standard basis vectors. In fact it turns out that $e1$ is not in the column space of A, so we get our $u4$ right away:

```
> e1 := vector([1,0,0,0]);
```

$$e1 := [1, 0, 0, 0]$$

```
> w4 := e1 - dotprod(u1,e1)*u1 - dotprod(u2,e1)*u2 -dotprod(u3,e1)*u3;
```

$$w4 := e1 - \frac{1}{6}\sqrt{6}\,u1 - \frac{1}{6}\sqrt{2}\,u2 - \frac{2}{3}\,u3$$

```
> evalm(w4);
```

$$\left[\frac{1}{3}, 0, \frac{-1}{3}, \frac{1}{3}\right]$$

```
> u_4 := normalize(w4);
```

$$u\_4 := \left[\frac{1}{3}\sqrt{3}, 0, -\frac{1}{3}\sqrt{3}, \frac{1}{3}\sqrt{3}\right]$$

Now, there was an easier way to do all of this. As expalined in the notes, if we organize the orthonrmal vectors $u1$, $u2$ and $u3$ into a matrix Q, there is another matrix R, which will be partially row-reduced, so that

$$A = QR\ .$$

You can find Q and R using the *QRdecomp( )* command of Maple. Using the command in the form

$$\mathtt{R} := \mathtt{QRdeconp}(\mathtt{A}, \mathtt{Q} =' \mathtt{Q}');$$

gives us the R factor, and then we can type

$$\mathtt{evalm}(\mathtt{Q})$$

to get the Q factor. We can then read off basis for the column space from Q. Let's try it:

```
> R := QRdecomp(A, Q='Q');
```

$$R := \begin{bmatrix} \sqrt{6} & \frac{3}{2}\sqrt{6} & \frac{7}{2}\sqrt{6} & \frac{1}{2}\sqrt{6} \\ 0 & \frac{3}{2}\sqrt{2} & \frac{3}{2}\sqrt{2} & \frac{1}{6}\sqrt{2} \\ 0 & 0 & 0 & \frac{2}{3} \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

```
> evalm(Q);
```

$$\begin{bmatrix} \frac{1}{6}\sqrt{6} & \frac{1}{6}\sqrt{2} & \frac{2}{3} & \frac{1}{3}\sqrt{3} \\ 0 & \frac{2}{3}\sqrt{2} & \frac{-1}{3} & 0 \\ \frac{1}{3}\sqrt{6} & 0 & 0 & -\frac{1}{3}\sqrt{3} \\ \frac{1}{6}\sqrt{6} & -\frac{1}{6}\sqrt{2} & \frac{-2}{3} & \frac{1}{3}\sqrt{3} \end{bmatrix}$$

First, let's check that $A = QR$:

```
> evalm(Q &* R);
```

$$\begin{bmatrix} 1 & 2 & 4 & 1 \\ 0 & 2 & 2 & 0 \\ 2 & 3 & 7 & 1 \\ 1 & 1 & 3 & 0 \end{bmatrix}$$

That checks out. Now notice that the first three columns of Q are the same $u1$, $u2$ and $u3$ that we found above. The fourth column is the u_4 that we found above. The `QRdecomp` command as we used it above always goes on to compute a full orthonormal basis for R^m when A is an m by n matrix. How do you know which ones come from the column space and which ones are "extra"? You look at the rows of R. Since the last row of R is all zeros, the last column of Q does not contribute to A. so the first three columns of Q are our orthonormal basis for the column space of A.

In general, if the first k rows of R are the non-zero rows of R, then the frist k columns of Q are the orthonormal basis for the column space of A. We will see what the others are a basis for in the next section when we discuss orthogonal complements**II** Now let's solve this one using the `QRdecomp` command. Don't

be alarmed if that is not the way you did it. You should know how to do it with pencil on paper. But it is also useful to know how to quickly check your work with Maple. In our solution to problem I, we wen't through all of the "pencil and paper" steps. Now let's go straight to the answer with Maple.

This time we will call the R part RB, and the Q part QB. These aren't products, just names using two capital letters.

```
> B := transpose(A);
```

$$B := \begin{bmatrix} 1 & 0 & 2 & 1 \\ 2 & 2 & 3 & 1 \\ 4 & 2 & 7 & 3 \\ 1 & 0 & 1 & 0 \end{bmatrix}$$

```
> RB := QRdecomp(A, Q='QB');
```

$$RB := \begin{bmatrix} \sqrt{6} & \frac{3}{2}\sqrt{6} & \frac{7}{2}\sqrt{6} & \frac{1}{2}\sqrt{6} \\ 0 & \frac{3}{2}\sqrt{2} & \frac{3}{2}\sqrt{2} & \frac{1}{6}\sqrt{2} \\ 0 & 0 & 0 & \frac{2}{3} \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

```
> evalm(QB);
```

$$QB := \begin{bmatrix} \frac{1}{6}\sqrt{6} & \frac{1}{6}\sqrt{2} & \frac{2}{3} & \frac{1}{3}\sqrt{3} \\ 0 & \frac{2}{3}\sqrt{2} & \frac{-1}{3} & 0 \\ \frac{1}{3}\sqrt{6} & 0 & 0 & -\frac{1}{3}\sqrt{3} \\ \frac{1}{6}\sqrt{6} & -\frac{1}{6}\sqrt{2} & \frac{-2}{3} & \frac{1}{3}\sqrt{3} \end{bmatrix}$$

The orthonormal basis in question consists of the first three columns of QB. Notice that written as rows, these would be an orthonormal basis for the row space of A.

III This is now easy to do using the `QRdecomp( )` command. Of course, on a quiz you would do it by hand, but it is useful to know how to check your answers using Maple.

```
> A := matrix([[1,2,3],[0,3,2],[2,0,1]]);
```

$$A := \begin{bmatrix} 1 & 2 & 3 \\ 0 & 3 & 2 \\ 2 & 0 & 1 \end{bmatrix}$$

```
> R := QRdecomp(A,Q='Q');
```

$$R := \begin{bmatrix} \sqrt{5} & \frac{2}{5}\sqrt{5} & \sqrt{5} \\ 0 & \frac{1}{5}\sqrt{305} & \frac{10}{61}\sqrt{305} \\ 0 & 0 & \frac{7}{61}\sqrt{61} \end{bmatrix}$$

```
> evalm(Q);
```

$$\begin{bmatrix} \frac{1}{5}\sqrt{5} & \frac{8}{305}\sqrt{305} & \frac{6}{61}\sqrt{61} \\ 0 & \frac{3}{61}\sqrt{305} & -\frac{4}{61}\sqrt{61} \\ \frac{2}{5}\sqrt{5} & -\frac{4}{305}\sqrt{305} & -\frac{3}{61}\sqrt{61} \end{bmatrix}$$

```
> evalm(Q &* R);
```

$$\begin{bmatrix} 1 & 2 & 3 \\ 0 & 3 & 2 \\ 2 & 0 & 1 \end{bmatrix}$$